

Math 262a, Fall 1999, Glenn Tesler

WZ method demo

10/29/99

```
> read 'hsum.mpl';
```

Copyright 1998 Wolfram Koepf, Konrad-Zuse-Zentrum Berlin

Example 1. Prove

```
> Sum(binomial(n,k),k)=2^n;
```

$$\sum_k \text{binomial}(n, k) = 2^n$$

Divide summand by right side

```
> F1 := (n,k) -> binomial(n,k) / 2^n;
```

$$F1 := (n, k) \rightarrow \frac{\text{binomial}(n, k)}{2^n}$$

Set up term for Gosper's algorithm

```
> a1 := F1(n+1,k) - F1(n,k);
```

$$a1 := \frac{\text{binomial}(n+1, k)}{2^{(n+1)}} - \frac{\text{binomial}(n, k)}{2^n}$$

```
> G1 := gosper(a1,k);
```

$$G1 := - \frac{k \left(\frac{\text{binomial}(n+1, k)}{2^{(n+1)}} - \frac{\text{binomial}(n, k)}{2^n} \right)}{-n-1+2k}$$

Check $F(n+1,k) - F(n,k) = G(n,k+1) - G(n,k)$.

To do this, divide both sides by $F(n,k)$; that makes both sides rational functions of n,k . Simplify them till they are in rational form. Then it's easy to compare them.

left side:

```
> lh1 := (F1(n+1,k) - F1(n,k)) / F1(n,k);
```

$$lh1 := \frac{\left(\frac{\text{binomial}(n+1, k)}{2^{(n+1)}} - \frac{\text{binomial}(n, k)}{2^n} \right) 2^n}{\text{binomial}(n, k)}$$

Maple's built-in simplification:

```
> simplify(lh1);
```

$$-\frac{1}{2} \frac{-n-1+2k}{k-1-n}$$

And if that had failed, Koepf has a fancier one designed for hypergeometric expressions:

```
> simpcomb(lh1);
```

$$-\frac{1}{2} \frac{-n-1+2k}{k-1-n}$$

right side:

(This is ugly... F1 was defined in Maple as a function of two variables, while G1 is just an algebraic expression, so parameters must be given to F1, and cannot be given to G1; but we want k replaced by k+1 in G1, so we use subs.)

```
> rh1 := (subs(k=k+1,G1) - G1)/F1(n,k);
```

$$rh1 := \left(-\frac{(k+1) \left(\frac{\text{binomial}(n+1, k+1)}{2^{(n+1)}} - \frac{\text{binomial}(n, k+1)}{2^n} \right)}{2k+1-n} + \frac{k \left(\frac{\text{binomial}(n+1, k)}{2^{(n+1)}} - \frac{\text{binomial}(n, k)}{2^n} \right)}{-n-1+2k} \right) 2^n / \text{binomial}(n, k)$$

```
> simplify(rh1);
```

$$-\frac{1}{2} \frac{-n-1+2k}{k-1-n}$$

```
> simplify(lh1-rh1);
```

$$0$$

Note: the comparison function doesn't work because it asks whether the expressions are coded identically, not whether they are algebraically equal. The specific formulas given for lh1, rh1 above, are different expressions, though algebraically they are equal.

```
> evalb(lh1=rh1);
```

false

```
> evalb((x+1)^2=x^2+2*x+1);
```

false

```
> (x+1)^2;
```

$$(x+1)^2$$

```
> x^2+2*x+1;
```

$$x^2 + 2x + 1$$

```
> 2*(x^2+1);
```

$$2x^2 + 2$$

Maple does some simplifications on its own, but only simple ones, so this test is true:

```
> evalb(2*(x^2+1)=2*x^2+2);
```

true

Now, back to WZ... The certificate is

```
> R1 := G1 / F1(n,k);
```

$$R1 := - \frac{k \left(\frac{\text{binomial}(n+1, k)}{2^{(n+1)}} - \frac{\text{binomial}(n, k)}{2^n} \right) 2^n}{(-n-1+2k) \text{binomial}(n, k)}$$

```
> R1 := simplify(R1);
```

$$R1 := \frac{1}{2} \frac{k}{k-1-n}$$

Now do the same problem over again with Koepf's single function that does much of the above:

```
> R1b := WZcertificate(F1(n,k), k, n);
```

$$R1b := \frac{1}{2} \frac{k}{k-1-n}$$

The fact that it succeeded implies $f1(n) = \sum_k F1(n,k)$ is a constant. By contrast,

```
> WZcertificate(binomial(n,k), k, n);
```

Error, (in WZcertificate) extended WZ method fails

Now, to prove our identity from the certificate, do this:

```
> G1b := F1(n,k) * R1b;
```

$$G1b := \frac{1}{2} \frac{\text{binomial}(n, k) k}{2^n (k-1-n)}$$

```
> simplify((F1(n+1,k) - F1(n,k))/F1(n,k));
```

$$-\frac{1}{2} \frac{-n-1+2k}{k-1-n}$$

```
> simplify((subs(k=k+1, G1b) - G1b)/F1(n,k));
```

$$-\frac{1}{2} \frac{-n-1+2k}{k-1-n}$$

```
> simplify(" "-");
```

$$0$$

The equation $F(n+1,k)-F(n,k) = G(n,k+1)-G(n,k)$ holds (where $F=F1$, $G=G1b$).

Thus $f1(n)$ is constant. To determine it,

```
> sum(F1(0,k), k=-infinity..infinity);
```

$$\sum_{k=-\infty}^{\infty} \text{binomial}(0, k)$$

It can't figure that out, so figure the support by hand and plug it in.

```
> sum(F1(0,k),k=0..0);
```

1

```
>
```

```
>
```

Companion identity:

Define a new function $g1(k) = \sum_{n \geq 0} G1(n,k)$:

(Using = instead of := makes it print nicely, but it doesn't actually assign a value to "g1(k)")

```
> g1(k) = Sum(G1,n=0..infinity);
```

$$g1(k) = \sum_{n=0}^{\infty} \left(- \frac{k \left(\frac{\text{binomial}(n+1, k)}{2^{(n+1)}} - \frac{\text{binomial}(n, k)}{2^n} \right)}{-n-1+2k} \right)$$

or

```
> simpG1 := simpcomb(G1):
```

```
g1 = Sum(simpG1,n=0..infinity);
```

$$g1 = \sum_{n=0}^{\infty} \left(- \frac{1}{2} \frac{\Gamma(n+1)}{\Gamma(k) \Gamma(n+2-k) 2^n} \right)$$

Note this is $-\text{binomial}(n,k-1)/2^{(n+1)}$:

```
> bG1 := -binomial(n,k-1)/2^(n+1):
```

```
convert(bG1,GAMMA);
```

$$- \frac{\Gamma(n+1)}{\Gamma(k) \Gamma(n+2-k) 2^{(n+1)}}$$

From $F1(n+1,k) - F1(n,k) = G1(n,k+1) - G1(n,k)$, we can obtain a telescoping relation in n on the left side. Sum it over $n=0,1,2,\dots$

So $F1(\text{infinity},k) - F1(0,k) = g1(k+1) - g1(k)$. In this case:

```
> F1(infinity,k);
```

$$\frac{\text{binomial}(\infty, k)}{2^{\infty}}$$

```
> limit(F1(n,k),n=infinity);
```

0

```
> F1(0,k);
```

$$\text{binomial}(0, k)$$

So for integers $k < 0$, we get $0 - 0 = g_1(k+1) - g_1(k)$,

and for $k=0$ we get $0 - 1 = g_1(1) - g_1(0)$.

Thus, for $k=1,2,3,4,\dots$ we have $g_1(k)=C$ for some constant C ,

and for $k=0,-1,-2,-3,-4,\dots$ we have $g_1(k)=\text{different constant}=C+1$:

```
> Sum(bG1,n=0..infinity) = piecewise(k>=1,C,C+1);
```

$$\sum_{n=0}^{\infty} \left(-\frac{\text{binomial}(n, k-1)}{2^{(n+1)}} \right) = \begin{cases} C & 1 \leq k \\ C+1 & \text{otherwise} \end{cases}$$

We will see that $C=-1$. Plugging this in, multiplying by -2 , and shifting k by 1 , will give the final answer

```
> Sum(binomial(n,k)/2^n,n=0..infinity) =
    piecewise(k>=0,2,k<0,0);
```

$$\sum_{n=0}^{\infty} \frac{\text{binomial}(n, k)}{2^n} = \begin{cases} 2 & 0 \leq k \\ 0 & k < 0 \end{cases}$$

```
>
```

Let's find C and check this empirically.

```
> g1 := sum(G1,n=0..infinity);
```

$$g_1 := - \frac{k \left(\frac{1}{2} \text{binomial}(1, k) - \text{binomial}(0, k) \right) \text{hypergeom}\left([1, 1], [2 - k], \frac{1}{2}\right)}{-1 + 2k}$$

Check it:

```
> for kk from -5 to 5 do
    val := subs(k=kk,g1):
    print(' g1\'(kk)=evalf(val));
od:
```

$$g_1(-5) = 0$$

$$g_1(-4) = 0$$

$$g_1(-3) = 0$$

$$g_1(-2) = 0$$

$$g_1(-1) = 0$$

$$g_1(0) = 0$$

$$g_1(1) = -1.$$

$$g_1(2) = 0$$

$$g_1(3) = 0$$

$$g_1(4) = 0$$

$$g_1(5) = 0$$

[Something is wrong... Maple is confused by singularities.

```
> subs(k=1,g1);
```

$$-\left(\frac{1}{2} \text{binomial}(1,1) - \text{binomial}(0,1)\right) \text{hypergeom}\left([1,1],[1],\frac{1}{2}\right)$$

```
> evalf(");
```

-1.

```
> subs(k=2,g1);
```

$$-\frac{2}{3}\left(\frac{1}{2} \text{binomial}(1,2) - \text{binomial}(0,2)\right) \text{hypergeom}\left([1,1],[0],\frac{1}{2}\right)$$

```
> evalf(");
```

0

[It evaluated binomial(1,2)=binomial(0,2)=0, and then never evaluated hypergeom(...) because it thought 0*something=0. Let's redo it.

```
> g1;
```

$$-\frac{k\left(\frac{1}{2} \text{binomial}(1,k) - \text{binomial}(0,k)\right) \text{hypergeom}\left([1,1],[2-k],\frac{1}{2}\right)}{-1+2k}$$

[Rewrite the hypergeom as an infinite sum over a new variable, r..

```
> glr := subs(hypergeom([1,1],[2-k],1/2) =
hyperterm([1,1],[2-k],1/2,r),g1);
```

$$glr := -\frac{k\left(\frac{1}{2} \text{binomial}(1,k) - \text{binomial}(0,k)\right) r! \left(\frac{1}{2}\right)^r}{(-1+2k) \text{pochhammer}(2-k,r)}$$

```
> glr := simpcomb(glr);
```

$$glr := -\frac{1}{2} \frac{\Gamma(r+1) \left(\frac{1}{2}\right)^r}{\Gamma(2-k+r) \Gamma(k)}$$

[Check that summing over r would give what we started with.

```
> glr2:=sumtools[sumtohyper](glr,r);
```

$$glr2 := \frac{1}{2} \frac{\sin(\pi k) \text{hypergeom}\left([1,1],[2-k],\frac{1}{2}\right)}{\pi(k-1)}$$

[Using the reflection formula for the Gamma function converts

```
> binomial(0,k) = convert(binomial(0,k),GAMMA);
```

$$\text{binomial}(0, k) = \frac{1}{\Gamma(k+1) \Gamma(1-k)}$$

```
> simplify(");
```

$$\text{binomial}(0, k) = -\frac{\sin(\pi(k+1))}{k\pi}$$

explaining what just happened. Now continue.

```
> for kk from -5 to 5 do
    glrk := limit(g1r, k=kk):
    print(' g1'(kk)=sum(g1rk, r=0..infinity));
od:
```

$$g1(-5) = 0$$

$$g1(-4) = 0$$

$$g1(-3) = 0$$

$$g1(-2) = 0$$

$$g1(-1) = 0$$

$$g1(0) = 0$$

$$g1(1) = -1$$

$$g1(2) = -1$$

$$g1(3) = -1$$

$$g1(4) = -1$$

$$g1(5) = -1$$

Note: We had to do it term by term, maple doesn't know how to take the limit otherwise:

```
> limit(g1, k=2);
```

$$\lim_{k \rightarrow 2} -\frac{k \left(\frac{1}{2} \text{binomial}(1, k) - \text{binomial}(0, k) \right) \text{hypergeom}\left([1, 1], [2-k], \frac{1}{2}\right)}{-1+2k}$$

```
> limit(g1r2, k=2);
```

$$\lim_{k \rightarrow 2} \frac{1}{2} \frac{\sin(\pi k) \text{hypergeom}\left([1, 1], [2-k], \frac{1}{2}\right)}{\pi(k-1)}$$

```
> limit(g1r2, k=-2);
```

$$\lim_{k \rightarrow (-2)} \frac{1}{2} \frac{\sin(\pi k) \operatorname{hypergeom}\left([1, 1], [2 - k], \frac{1}{2}\right)}{\pi(k - 1)}$$

Example 2. Gauss's 2F1:

> `Hypergeom([a,b],[c],1) =`
`pochhammer(c-b,-a)/pochhammer(c,-a);`

$$\operatorname{Hypergeom}([a, b], [c], 1) = \frac{\operatorname{pochhammer}(c - b, -a)}{\operatorname{pochhammer}(c, -a)}$$

which also can be written

> `convert(rhs("), GAMMA);`

$$\frac{\Gamma(c - b - a) \Gamma(c)}{\Gamma(c - b) \Gamma(c - a)}$$

There's no summation symbol in the above, but remember the hypergeometric series notation abbreviates an infinite sum:

> `F2a := hyperterm([a,b],[c],1,k);`

$$F2a := \frac{\operatorname{pochhammer}(a, k) \operatorname{pochhammer}(b, k)}{\operatorname{pochhammer}(c, k) k!}$$

> `r2 := pochhammer(c-b,-a)/pochhammer(c,-a);`

$$r2 := \frac{\operatorname{pochhammer}(c - b, -a)}{\operatorname{pochhammer}(c, -a)}$$

> `F2b := F2a/r2;`

$$F2b := \frac{\operatorname{pochhammer}(a, k) \operatorname{pochhammer}(b, k) \operatorname{pochhammer}(c, -a)}{\operatorname{pochhammer}(c, k) k! \operatorname{pochhammer}(c - b, -a)}$$

Also, there's no "n" in the above. Let a=-n.

We want to prove $f2(n) = \sum_k F2(n,k)$ equals 1.

> `F2c := subs(a=-n,F2b);`

$$F2c := \frac{\operatorname{pochhammer}(-n, k) \operatorname{pochhammer}(b, k) \operatorname{pochhammer}(c, n)}{\operatorname{pochhammer}(c, k) k! \operatorname{pochhammer}(c - b, n)}$$

> `R2 := WZcertificate(F2c,k,n);`

$$R2 := \frac{(c + k - 1) k}{(k - 1 - n) (c - b + n)}$$

This proves the original sum is true for integer n, i.e., integer a. Gauss didn't have the restriction that it must be an integer. Here's a way around it..


```

> F2d := subs(a=-n-eps, F2b);
      F2d := 
$$\frac{\text{pochhammer}(-n - \text{eps}, k) \text{pochhammer}(b, k) \text{pochhammer}(c, n + \text{eps})}{\text{pochhammer}(c, k) k! \text{pochhammer}(c - b, n + \text{eps})}$$

> R2d := WZcertificate(F2d, k, n);
      R2d := 
$$\frac{(c + k - 1) k}{(-n - 1 - \text{eps} + k) (c - b + n + \text{eps})}$$

eps, b, c are constants that were just carried along for the ride. For all integer n,
Gauss's 2F1 sum holds for arbitrary b, c and for a=-n-eps, where eps is also arbitrary;
thus, a is arbitrary.
If we want, we can verify the proof:
> G2d := F2d * R2d;
G2d := 
$$\frac{\text{pochhammer}(-n - \text{eps}, k) \text{pochhammer}(b, k) \text{pochhammer}(c, n + \text{eps})}{(c + k - 1) k / (\text{pochhammer}(c, k) k! \text{pochhammer}(c - b, n + \text{eps}) (-n - 1 - \text{eps} + k) (c - b + n + \text{eps}))}$$

> (subs(n=n+1, F2d) - F2d) / F2d = (subs(k=k+1, G2d) - G2d) / F2d;

$$\left( \frac{\text{pochhammer}(-n - 1 - \text{eps}, k) \text{pochhammer}(b, k) \text{pochhammer}(c, n + 1 + \text{eps})}{\text{pochhammer}(c, k) k! \text{pochhammer}(c - b, n + 1 + \text{eps})} - \frac{\%3 \text{pochhammer}(b, k) \%1}{\text{pochhammer}(c, k) k! \%2} \right) \text{pochhammer}(c, k) k! \%2 / (\%3 \text{pochhammer}(b, k) \%1) =$$


$$\left( \frac{\text{pochhammer}(-n - \text{eps}, k + 1) \text{pochhammer}(b, k + 1) \%1 (c + k) (k + 1)}{\text{pochhammer}(c, k + 1) (k + 1)! \%2 (-n - \text{eps} + k) (c - b + n + \text{eps})} - \frac{\%3 \text{pochhammer}(b, k) \%1 (c + k - 1) k}{\text{pochhammer}(c, k) k! \%2 (-n - 1 - \text{eps} + k) (c - b + n + \text{eps})} \right) \text{pochhammer}(c, k) k! \%2 / (\%3 \text{pochhammer}(b, k) \%1)$$

%1 := pochhammer(c, n + eps)
%2 := pochhammer(c - b, n + eps)
%3 := pochhammer(-n - eps, k)
> simplify(");

$$- \frac{n b + b + \text{eps} b + k c - k b + n k + k \text{eps}}{(-n - 1 - \text{eps} + k) (c - b + n + \text{eps})} =$$


$$- \frac{n b + b + \text{eps} b + k c - k b + n k + k \text{eps}}{(-n - 1 - \text{eps} + k) (c - b + n + \text{eps})}$$

> evalb(");
true

```

See the tables on Koepf pp. 84, 87 for a lot of other identities that are proved in this same fashion.

Example 3. A=B Example 7.3.1

> F3 := binomial(n,k)^2/binomial(2*n,n);

$$F3 := \frac{\text{binomial}(n, k)^2}{\text{binomial}(2n, n)}$$

> R3 := WZcertificate(F3,k,n);

$$R3 := \frac{1}{2} \frac{(-3n - 3 + 2k)k^2}{(2n + 1)(k - 1 - n)^2}$$

So this implies that for n=0,1,2,...,

> Sum(F3,k)=constant;

$$\sum_k \frac{\text{binomial}(n, k)^2}{\text{binomial}(2n, n)} = \text{constant}$$

Evaluate the constant.

> sum(subs(n=0..infinity, F3), k=0..0);

1

Thus, the above sum equals 1, so this proves

> Sum(binomial(n,k)^2,k)=binomial(2*n,n);

$$\sum_k \text{binomial}(n, k)^2 = \text{binomial}(2n, n)$$

Companion identity

> G3 := R3*F3;

$$G3 := \frac{1}{2} \frac{(-3n - 3 + 2k)k^2 \text{binomial}(n, k)^2}{(2n + 1)(k - 1 - n)^2 \text{binomial}(2n, n)}$$

Define

> g3(k) = Sum(G3,n=0..infinity);

$$g3(k) = \sum_{n=0}^{\infty} \left(\frac{1}{2} \frac{(-3n - 3 + 2k)k^2 \text{binomial}(n, k)^2}{(2n + 1)(k - 1 - n)^2 \text{binomial}(2n, n)} \right)$$

Summing F(n+1,k)-F(n,k) = G(n,k+1)-G(n,k) for n=0..infinity gives

> 'F3(infinity,k)' - 'F3(0,k)' = g3(k+1)-g3(k);

$$F3(\infty, k) - F3(0, k) = g3(k + 1) - g3(k)$$

> subs(n=infinity, F3);

```
[
    
$$\frac{\text{binomial}(\infty, k)^2}{\text{binomial}(\infty, \infty)}$$

    > limit(F3,n=infinity);
    0
    > subs(n=0,F3);
    
$$\frac{\text{binomial}(0, k)^2}{\text{binomial}(0, 0)}$$

]
```

which is 1 if k=0 and is 0 otherwise, i.e., delta(0,k). So this proves for integer k,

```
> g3(k+1)-g3(k) = piecewise(k<>0,0-0,k=0,0-1);
```

$$g3(k+1) - g3(k) = \begin{cases} 0 & k \neq 0 \\ -1 & k = 0 \end{cases}$$

Now evaluate at any k.

```
> sum(subs(k=0,G3),n=0..infinity);
0
```

This proves g3(0)=0 so g3(1)=-1. Thus

```
> g3(k) = piecewise(k>=1,-1,k<=0,0);
```

$$g3(k) = \begin{cases} -1 & 1 \leq k \\ 0 & k \leq 0 \end{cases}$$

i.e.,

```
> Sum(G3,n=0..infinity)=rhs(");
```

$$\sum_{n=0}^{\infty} \left(\frac{1}{2} \frac{(-3n-3+2k)k^2 \text{binomial}(n,k)^2}{(2n+1)(k-1-n)^2 \text{binomial}(2n,n)} \right) = \begin{cases} -1 & 1 \leq k \\ 0 & k \leq 0 \end{cases}$$

which can be rearranged into

```
> Sum(convert(simpcomb(-2*subs(k=k+1,G3)),binomial),n=0..infinity) = piecewise(k>=0,2,k<=-1,0);
```

$$\sum_{n=0}^{\infty} \frac{(3n+1-2k) \text{binomial}(n,k)^2}{(n+1) \text{binomial}(2n+1,n)} = \begin{cases} 2 & 0 \leq k \\ 0 & k \leq -1 \end{cases}$$

```
>
```

Example 4. Identities where the WZ method fails

```
> F4 := (n,k) -> (-1)^k / (-3)^n * binomial(n,k) *
    binomial(3*k,n);
f4 := n -> sum(F4(n,k),k=0..n); # interval 0..n contains
support
```

$$F4 := (n, k) \rightarrow \frac{(-1)^k \text{binomial}(n, k) \text{binomial}(3k, n)}{(-3)^n}$$

$$f4 := n \rightarrow \sum_{k=0}^n F4(n, k)$$

```
> WZcertificate(F4(n,k),k,n);
Error, (in WZcertificate) extended WZ method fails
```

But the function really is constant:

```
> 'f4(nn)' $nn=0..15;
```

1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1

Find a recursion it does satisfy:

```
> sumrecursion(F4(n,k),k,s(n));
```

$$2(2n+3)s(n+2) - (5n+7)s(n+1) + (n+1)s(n) = 0$$

Thus, $LF = 0$ where $L = L4 = 2(2n+3)E^2 - (5n+7)E + (n+1)$ and $F = F4$.

```
> L4 := 2*(2*n+3)*E^2 - (5*n+7)*E + (n+1);
```

$$L4 := 2(2n+3)E^2 - (5n+7)E + n + 1$$

The software we're using now doesn't handle this noncommutative algebra, but we can "fake" it in this case: if we systematically keep n 's on the left and E 's on the right, then any factorization of the form

$$(*) \quad p(n) * q(n, E) * r(E)$$

(where within q , the n 's are on the left and E 's are on the right) results in the same computation whether it's done commutatively or noncommutatively. Do the normal commutative factorization:

```
> factor(L4);
```

$$(E-1)(4En+6E-n-1)$$

This expression is literally false. As far as maple is concerned, $n*E=E*n$, although in reality, $(n+1)*E=E*n$. But we can make sense of it: since this was treated as a commutative polynomial, Maple thinks it's the same as

$$L4 = ((4n-6)E - (n+1)) * (E-1)$$

and this has the form $(*)$ given above. This is a valid factorization. So

$$((4*n-6)*E - (n+1)) * (E-1) * 'f4(n)' = 0;$$

$$((4n-6)E - n - 1)(E-1)f4(n) = 0$$

is true. Now we want to prove that in fact $(E-1)f4(n) = 0$, which is a right factor of this whole mess. Let $h4(n) = (E-1)f4(n)$.

```
> h4 := n -> f4(n+1) - f4(n);
```

$$h4 := n \rightarrow f4(n+1) - f4(n)$$

Then $((4n-6)E - (n+1))h4(n) = 0$. This has order 1, so it has a solution $A * h4a(n)$ for some function $h4a(n)$, and the constant A is determined by one initial value:

```
> h4(0); 'h4(nn)' $'nn'=0..10;
```

0

0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0

[Observe that $h_4(n)=0$ satisfies the recursion equation and the initial value, and therefore the sole solution is $h_4(n) = 0$ identically. So (E-1) $f_4(n)=0$, so $f_4(n+1)=f_4(n)$, as we wanted.

[>

[>